


# TIPS 'N TECHNIQUES



## Applesoft Windows

by Michael A. Seeds, Franklin and Marshall College, P.O. Box 3003, Lancaster, PA 17604

Your Apple Monitor contains a window, and looking through that window can solve a few bothersome programming problems. For example, I like to jump around when I am editing a program, and sometimes I need to copy parts of one section into another section. I've often wished I could run two video monitors side by side — one to display the program and one to display my working area. Another problem is displaying short messages without disturbing text already on the screen.

We can solve problems like these using the Apple text window. Normally the window is set to fill the entire video screen, but you can change the boundaries of the text window by POKEing locations 32 to 35. A POKE 32,10, for instance, sets the left margin of the text window to the tenth space from the left. You can try this in immediate mode, if you like. Try these locations:

Location	Function	Limits
32	Left edge	0-39
33	Width of window	1-40
34	Top edge	0-23
35	Bottom edge	0-23

**DOS 3.3** *The Apple's text display is a window that you can control from within your program. These short programs show you how it's done in both Applesoft and machine language.* **ProDOS**

(For more details, see page 31 of the *Apple II Reference Manual* or Appendix F of the; Applesoft BASIC Programmer's Reference Manual.)

When you change the text window, your Apple uses the new area and ignores anything outside it. The HOME command clears just the window and places the cursor at the upper left corner. If you list a program, the text window will scroll as usual, but text outside the window will be left untouched. HTAB will move the cursor relative to the newly defined left edge, but VTAB will allow you to move the cursor above or below the existing text window. Go ahead and try a few POKES. No matter how much you mess up the window boundaries, you can restore the window to full screen with a TEXT command.

### PROGRAM CONTROL OF WINDOWS

Using these same locations, programs can very easily control the window boundaries.

The following two programs present two possibilities. The first, BIWIND, gives you two video work areas for program development and testing, while the second, WINDER, demonstrates a technique for creating Macintosh-like "dialog" windows. BIWIND is a short machine language program that allows you to divide your video screen into a top screen and a bottom screen. You can work in either half without disturbing the other half. With BIWIND installed, you can enter the top half of the screen by moving the cursor to the bottom half and typing CALL 771. When you press <RETURN> the screen will divide in half and you will be working in the top half. You can LIST, EDIT, and RUN programs here without disturbing the text below (unless, of course, your program alters locations 32 to 35 or uses a TEXT command). To enter the bottom half of the screen, move the cursor to the top half and type CALL 794. Press <RETURN> and the bottom window will open for your use. To return to the full screen, just type TEXT.

The program will always divide the screen into two equal areas unless you specify otherwise. To divide the screen at the nth line, just POKE 770, n. The next time you open one of the work areas, the new dividing line will be in effect.

## ENTERING THE PROGRAM

To key in BIWIND, enter the machine language code shown in Listing 1 and save it on disk with the command:

### BSAVE BIWIND,A\$300,L\$40

For help in entering *Nibble* listings, see "A Welcome to New *Nibble* Readers" in the beginning of this issue.

## HOW THE PROGRAM WORKS

This machine language program is really two separate routines. The first opens the top of the screen, while the second opens the bottom of the screen. Let's look at the first routine. When we CALL 771, the program saves the present location of the cursor for use when we flip back to the bottom half of the screen, and then it loads in the last location of the cursor in the top half of the screen. The Apple always stores the current cursor location in \$25. Next, the program sets the top of the working area to zero and

"The subroutines can be used in any BASIC program to open a small window in a text screen display."

it sets the bottom line to the contents of \$302. Finally, it calls the subroutine at \$334 to draw a line of equal signs dividing the two screen areas.

The second routine opens the bottom of the screen. This routine is very similar to the first. The major difference is that it sets the top of the screen to the contents of \$302 plus 1. This prevents it from overwriting the dividing line of equal signs.

BIWIND is a simple program, and, because of the way it remembers the last cursor position, it can get confused if you try to open the half of the screen in which you are already working. If that happens, try typing HOME. If things get hopelessly confused, just type TEXT and you will be back to a full screen.

I find BIWIND especially useful for editing my programs. For example, I can jump to the top screen to list one segment of the program, and then jump back to the bottom screen to edit a related part of the program.

The program WINDER (Listing 2) also uses the text window, but for a different purpose. The subroutines starting at line 390 can be used in any BASIC program to open

a small window in a text screen display. I use this to display messages to the user. The subroutine at line 560 closes the temporary message window and restores the original data. The first part of the program in Listing 2 is just a demonstration of the methods.

To open a window, the main program must set the quantities WL, WT, WW, and WB. These are the four numbers to be POKEd into locations 32 to 35, and they define the location and size of the window. To allow room for a border, WW and WB must be greater than two. Of course, the boundaries of the window must not go beyond the boundaries of the video screen.

## Entering the program

To key in WINDER, simply enter the Applesoft program shown in Listing 2 and save it on disk with the command:

### SAVE WINDER

If you decide to use these subroutines in your own programs, notice that the variables they use all begin with a W. Avoid using W variables in the rest of your program so that the window subroutines won't interfere with them. Notice, also, that your main program must dimension WSS(24).

If you follow these few rules, it's simple to open windows into your computer.

## LISTING 1: BIWIND

```

0
1
2
3
4
5
6
7
8
9
10
11
12
13
14 0300 0B
15 0301 17
16 0302 0C
17 0303 A5 25
18 0305 8D 01 03
19 0308 AD 00 03
20 030B 85 25
21 030D A9 00
22 030F 85 22
23 0311 AD 02 03
24 0314 85 23
25 0316 20 34 03
26 0319 60
27
28 031A A5 25
29 031C 8D 00 03
30 031F AD 01 03
31 0322 85 25
32 0324 AD 02 03
33 0327 18
34 0328 69 01
35 032A 85 22
36 032C A9 18
37 032E 85 23
38 0330 20 34 03
39 0333 60
40
41 0334 AD 02 03
42 0337 20 24 FC
43 033A A2 27
44 033C A9 BD
45 033E 20 F0 FD OUT
46 0341 CA
47 0342 D0 FA
48 0344 60

                                BIWIND
                                BY MIKE SEEDS
                                COPYRIGHT (C) 1985
                                BY MICROSPARC, INC.
                                CONCORD, MA 01742

                                MICROSPARC ASSEMBLER

                                ORG $300
                                EQU $25
                                EQU $23
                                EQU $22
                                DFC 11
                                DFC 12
                                DFC 12
                                LDA VCURS
                                STA BOT
                                LDA TOP
                                STA VCURS
                                LDA #0
                                STA TOPSCR
                                LDA LINE
                                STA BOTSCR
                                JSR DIVIDE
                                RTS

                                VERTICAL CURSOR POSITION
                                BOTTOM OF TEXT WINDOW
                                TOP EDGE OF WINDOW
                                TOP CURSOR POSITION
                                BOTTOM CURSOR POSITION
                                DIVIDING LINE SET AT 12
                                *** OPEN TOP ***
                                SAVE BOTTOM CURSOR POSITION
                                SET TOP CURSOR POSITION
                                SET TOP OF AREA
                                GET DIVIDING LINE
                                SET BOTTOM OF AREA
                                DRAW DIVIDING LINE
                                END OF OPEN TOP ROUTINE

                                *** OPEN BOTTOM ***
                                SAVE TOP CURSOR POSITION
                                SET BOTTOM CURSOR POSITION
                                GET DIVIDING LINE
                                ADD ONE
                                SET TOP OF AREA
                                DECIMAL 24
                                SET BOTTOM OF AREA
                                DRAW DIVIDING LINE
                                END OF OPEN BOTTOM ROUTINE

                                DIVIDE LDA LINE
                                JSR $FC24
                                LDX #39
                                LDA #SBD
                                JSR $FDF0
                                DEX
                                BNE OUT
                                RTS

                                GET LINE POSITION
                                VTAB TO DIVIDING LINE
                                PRINT 39 SYMBOLS
                                = SIGN
                                PRINT A SYMBOL
                                LAST SYMBOL?
                                DONE

000 ERRORS
0300 HEX START OF OBJECT
0344 HEX END OF OBJECT
0045 HEX LENGTH OF OBJECT
95C3 HEX END OF SYMBOLS
END OF LISTING 1
```

## LISTING 2: WINDER

```

10 REM *****
20 REM * WINDER *
30 REM * BY MIKE SEEDS *
40 REM * COPYRIGHT (C) 1985 *
50 REM * BY MICROSPARC, INC *
60 REM * CONCORD, MA 01742 *
70 REM *****
80 DIM WS$(24)
90 HOME : PRINT : PRINT TAB( 9)"FOR WHOM TH
   E BELL BONGS"
100 PRINT : PRINT TAB( 15)"BY A. MONKEY": PRINT
   : PRINT
110 FOR J = 1 TO 8
120 FOR K = 1 TO 40: PRINT CHR$( 64 + 26 *
   RND (1));: NEXT K: PRINT
130 NEXT J
140 VTAB 23: PRINT "PRESS ANY KEY TO HALT."
150 WL = 12:WT = 10:WW = 10:WB = 5: GOSUB 390
   : REM OPEN WINDOW
160 PRINT : PRINT "GOT IT?"
170 GOSUB 350: REM DELAY
180 GOSUB 560: REM CLOSE WINDOW
190 IF PEEK (49152) > 128 THEN TEXT : HOME
   : END
200 WL = 5:WT = 1:WW = 25:WB = 7
210 GOSUB 390: REM OPEN WINDOW
220 VTAB WT + 2: HTAB 4: PRINT "NOTICE THE T
   EXT IS": HTAB 4: PRINT "RESTORED CORRECT
   LY."
230 GOSUB 350: REM DELAY
240 GOSUB 560: REM CLOSE WINDOW
250 IF PEEK (49152) > 128 THEN TEXT : HOME
   : END
260 WT = 10:WB = 10: GOSUB 390
270 FOR J = 1 TO 25: PRINT " ";J,J * J: NEXT
   J

```

```

280 PRINT : PRINT "SCROLLING IS AUTOMATIC"
290 GOSUB 350: GOSUB 560
300 IF PEEK (49152) > 128 THEN TEXT : HOME
   : END
310 GOTO 150
320 REM =====
330 REM DELAY
340 REM =====
350 FOR J = 1 TO 1500: NEXT : RETURN
360 REM =====
370 REM SUBROUTINE WINDOW
380 REM =====
390 WA = 1024 + 128 * (WT - 1 - 8 * INT ((WT
   - 1) / 8)) + 40 * INT (WT / 8.5)
400 WS = WA
410 FOR WJ = WT TO WT + WB - 1:WS$(WJ) = ""
420 FOR WK = 1 TO WW:WS$(WJ) = WS$(WJ) + CHR$(
   ( PEEK (WA + WL + WK - 1)): NEXT WK
430 POKE WA + WL,32: POKE WA + WL + WW - 1,3
   2
440 WA = WA + 128: IF WA = 2088 THEN WA = 110
   4
450 IF WA = 2048 THEN WA = 1064
460 NEXT WJ
470 FOR WJ = 1 TO WW: POKE WS + WL + WJ - 1,
   32: POKE WA - 128 + 984 * (WA = 1064 OR
   WA = 1104) + WL + WJ - 1,32: NEXT WJ
480 REM SET TEXT SCREEN
490 POKE 32,WL + 1: POKE 33,WW - 2
500 POKE 34,WT: POKE 35,WT + WB - 2
510 HOME
520 RETURN
530 REM =====
540 REM SUBROUTINE CLOSE
550 REM =====
560 POKE 32,0: POKE 33,40
570 POKE 34,0: POKE 35,24
580 FOR WJ = WT TO WT + WB - 1: VTAB WJ: HTAB
   WL + 1: PRINT WS$(WJ): NEXT WJ
590 RETURN

```

END OF LISTING 2